Exercice n°2.

Couplage entre deux oscillateurs couplés

Soit deux pendules simples identiques (longueur L , masse m) , reliés par l'intermédiaire d'un ressort de raideur k.
Dans cette étude, les déplacements x_1 et x_2 , des points matériels M_1 et M_2 sont supposés faible par rapport à la longueur L.
Le ressort est au repos lorsque les fils des pendules sont verticaux (fig n°1).
Les frottements seront négligés dans la recherche du mouvement de M_1 et de M_2.



1-Etude du cas particulier : M_2 fixe et M_1 en mouvement.
M_2 est maintenu dans la position O'_2 (O_2,O'_2 vertical).
Le premier pendule est écarté d'un angle $\alpha_1(0)$de $\pi/4$ à l'instant t=0 puis relaché sans vitesse initiale.

11-ETABLIR l'équation différentielle satisfaite par $\alpha_1(t)$.

$$\boxed{\frac{d^2\alpha_1(t)}{dt^2} + \omega_0{}^2\alpha_1(t) = 0} \quad \text{avec} \quad \boxed{\omega o = \sqrt{\frac{g}{l} + \frac{k}{m}}}$$

Rappel : $I=mL^2$ (moment d'inertie du pendule).

12-TRACER l'évolution de $\alpha_1(t)$ en fonction du temps.

12-EN DEDUIRE la pulsation propre du mouvement.

2-Cas général.
A l'instant t=0 , les pendules sont déviés respectivement de $\alpha_1(0)$ et $\alpha_2(0)$ et lachés sans vitesse initiale.

21-A l'aide du théorème du moment cinétique , ETABLIR les équations différentielles dites couplées vérifiées par $\alpha_1(t)$ et $\alpha_2(t)$, en faisant intervenir $\Omega o=\sqrt{\frac{k}{m}}$

$$\boxed{\frac{d^2\alpha_1(t)}{dt^2} + \omega_0{}^2\alpha_1(t) - \Omega_0{}^2\alpha_2(t) = 0} \text{ équa}\mathbf{1} \text{ et } \boxed{\frac{d^2\alpha_2(t)}{dt^2} + \omega_0{}^2\alpha_2(t) - \Omega_0{}^2\alpha_1(t) = 0} \text{ équa } \mathbf{2}$$

22-RESOUDRE le système en utilisant $S(t)=\alpha_1(t)+\alpha_2(t)$ et $D(t)= \alpha_1(t)-\alpha_2(t)$
DONNER les expressions de $\alpha_1(t)$ et $\alpha_2(t)$ en tenant compte des conditions initiales.

équa**1-** équa **2** $\boxed{\frac{d^2S(t)}{dt^2} + (\omega_0{}^2 - \Omega_0{}^2)S(t) = 0}$

équa**1+** équa **2** $\boxed{\frac{d^2D(t)}{dt^2} + (\omega_0{}^2 - \Omega_0{}^2)D(t) = 0}$

23-TRACER l'évolution de $\alpha_1(t)$ et $\alpha_2(t)$ en fonction du temps.

Travail demandé : Ecrire un programme permettant de résoudre cette équation différentielle et tracer l'évolution de $\alpha_1(t)$ et $\alpha_2(t)$.

```
// pendules couplés avec ode
// --------------------------------------------------------
// système différentiel du second ordre
// les variables sont des vecteurs colonnes à 4 lignes
function z=Fpend(t, y)
l=1; // longueur du pendule
g=9.81 // constante de la gravité
k=1 //constante du ressort
m=1 //masse d'un pendule
omega=sqrt(g/l+k/m);
OMEGA=sqrt(k/m);
A=[0,0,1,0;0,0,0,1;-omega^2,OMEGA^2,0,0;OMEGA^2,-omega^2,0,0];
z=A*y;
// alternative :
// z =[y(3,1); y(4,1); -4*y(1,1)-4*y(2,1); 4*y(1,1)-4*y(2,1)];
// utile si le système est non linéaire par exemple
endfunction
// --------------------------------------------------------
t0=0;
Y0=[%pi/4; 0; 0; 0]; // instant t0 alpha1=%pi/4
// --------------------------------------------------------
T=10;
N=500;
h=T/N;
t0=0;
t=t0:h:t0+T; // t varie entre t0 et t0+T
Y=ode(Y0,t0,t,Fpend);
// --------------------------------------------------------
clf(0);
figure(0);
xtitle('angles en fonction du temps','$t$','$\alpha(t)$');
plot2d(t,Y(1,:)); // affiche le alpha1 en fonction de t
plot2d(t,Y(2,:)); // affiche le alpha2 en fonction de t
```



Remarque :

$$\begin{pmatrix} \dot{\alpha}_1 \\ \dot{\alpha}_2 \\ \ddot{\alpha}_1 \\ \ddot{\alpha}_2 \end{pmatrix} = \begin{pmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{21} & a_{22} & a_{23} & a_{24} \\ a_{31} & a_{32} & a_{33} & a_{34} \\ a_{41} & a_{42} & a_{43} & a_{44} \end{pmatrix} \times \begin{pmatrix} \alpha_1 \\ \alpha_2 \\ \dot{\alpha}_1 \\ \dot{\alpha}_2 \end{pmatrix}$$